In and Out of a Schwarzschild Black Hole

Frank Wang, fwang@lagcc.cuny.edu

In classical general relativity, an astronaut can cross the event horizon of a black hole without noticing it, while an external observer would witness the astronaut approaching but never reaching the event horizon as time tends to infinity. We integrate the differential equations of motion to show the difference between proper and coordinate times for a particle free falling into a Schwarzschild black hole, and present the particle world line in Kruskal and Penrose diagrams.

Roger Penrose received the 2020 Nobel Prize in Physics for his discovery that "black hole formation is a robust prediction of the general theory of relativity." He made the discovery using his innovative visualization of the spacetime known as the Penrose diagram.

Consider the Schwarzschild metric.

$$ds^2 = -\left(1 - \frac{2GM}{r}\right) dt^2 + \left(1 - \frac{2GM}{r}\right)^{-1} dr^2 + r^2 d\Omega$$

See Misner, Thorne, and Wheeler (MTW), Eq. (25.12). By setting G to unity we express t and r in geometrized units. We denote the Schwarzschild radius 2*M by rs to simplify the expressions. For a direct plunge, the angular part of the metric vanishes. The equations of motion are the following.

$$\frac{dt}{d\tau} = \frac{\tilde{E}}{1 - r_s/r}, \quad \frac{dr}{d\tau} = -\left(\tilde{E}^2 - 1 + \frac{r_s}{r}\right)^{1/2}$$

The derivatives are with respect to the proper time tau, and E-tilde is energy per mass, see MTW Eqs. (25.16) and (25.18). For a free fall from infinity, tilde-E is 1, and the equations become

$$\frac{dt}{d\tau} = \frac{1}{1 - r_s/r}, \quad \frac{dr}{d\tau} = -\left(\frac{r_s}{r}\right)^{1/2}$$

```
> restart:
```

The radial equation can be easily integrated.

```
> epr1 := Int(1/sqrt((rs/r)), r);
```

$$epr1 := \int \frac{1}{\sqrt{\frac{rs}{r}}} \, dr \tag{1}$$

```
> tau := -value(epr1);
```

$$\tau := -\frac{2r}{3\sqrt{\frac{rs}{r}}} \tag{2}$$

The above is the first part of MTW's Eq. (25.38).

We use the chain rule to obtain the following equation.

$$\frac{dr}{dt} = \frac{dr}{d\tau}\frac{d\tau}{dt}, \quad \frac{dr}{dt} = -\left(\frac{r_s}{r}\right)^{1/2}\left(1 - \frac{r_s}{r}\right)$$

```
> epr2 := Int(1/(sqrt(rs/r)*(1-rs/r)), r);
```

$$epr2 := \int \frac{1}{\sqrt{\frac{rs}{r}}\left(1 - \frac{rs}{r}\right)} \, dr \tag{3}$$

```
> tsch := - value(epr2);
```

(4)

$$tsch := \frac{2\left(3\,rs^{3/2}\operatorname{arctanh}\left(\frac{\sqrt{r}}{\sqrt{rs}}\right) - r^{3/2} - 3\,rs\sqrt{r}\right)}{3\sqrt{\frac{rs}{r}}\sqrt{r}} \qquad \textbf{(4)}$$

```
> tsch := convert(tsch, ln);
```

$$tsch := \frac{2\left(3\,rs^{3/2}\left(\frac{\ln\left(\frac{\sqrt{r}}{\sqrt{rs}}+1\right)}{2} - \frac{\ln\left(1-\frac{\sqrt{r}}{\sqrt{rs}}\right)}{2}\right) - r^{3/2} - 3\,rs\sqrt{r}\right)}{3\sqrt{\frac{rs}{r}}\sqrt{r}} \qquad \textbf{(5)}$$

The above is the second part of MTW's Eq. (25.38).
Before we plot the functions, we make a technical remark. Let's see the graph of the integral of 1/x.

```
> plot(int(1/x, x), x=-5..5);
```



But the integral of 1/x is actually log(abs(x)), not just log(x). Let's see what Maple tells us about negative logarithm.

```
> log(-1); log(-2);
```

$$I\pi$$

$$\ln(2) + I\pi \qquad \textbf{(6)}$$

From the above, we recall the fact log(-1)=I*Pi, which we have learned from complex analysis. Then log(-x) can be considered I*Pi+log(x). With this understanding, we take the real part of the integral to obtain the absolute value of x under the logarithm, the desired result.

```
> plot(Re(int(1/x, x)), x=-5..5);
```




```
> rs := 2;
```

$$rs := 2 \qquad (7)$$

The graph below shows that the proper time tau is well defined near the event horizon r=2.

```
> plot(tau, r=0..10, labels = [r, "proper time"]);
```



For the Schwarzschild coordinate t, the is a discontinuity at r=2, the event horizon indicated by the dashed line. Compare this with MTW Figure 31.1. Later we will plot the world line in different coordinates to eliminate the discontinuity.

```
> with(plots):
> pct := plot(Re(tsch), r=0..10, axes = boxed, labels = [r,
  "Schwarzschild time"]):
> peh := plot([rs, t, t=-25..10], linestyle = dash):
> display([pct, peh]);
```



We produce a plot similar to MTW's Figure 25.5.

```
> R := 6;
```

$$R := 6 \tag{8}$$

```
> ptau := plot([tau-eval(tau, r=R), r, r=0..R]):
> ptsch := plot([tsch-eval(tsch, r=R), r, r=0..R], color = blue):
> display([ptau, ptsch], labels = [time, r]);
```



In MTW's Figure 25.5 the energy tilde-E is sqrt(1-2*M/6), and here we use tilde-E=1. For the astronaut, the singularity r=0 is attained quickly---6.93 geometrized units to be precise.

```
> evalf(-eval(tau, r=R));
```

$$6.928203230 \tag{9}$$

For a faraway observer, r=0 is never reached, and r=2*M is attained asymptotically. In other words, one has to wait forever to see the astronaut touch the event horizon.

In Schwarzschild coordinates, r=2*M appears to be a singularity but it is not a physical singularity. We use the Kruskal-Szekeres coordinates for the world line to illustrate it. The relationship between Kruskal-Szekeres coordinates and Schwarzschild coordinates can be found in MTW Eq. (31.13).

```
> V1 := (t, r) -> sqrt(r/rs-1)*exp(r/2/rs)*sinh(t/2/rs);
```

$$V1 := (t, r) \mapsto \sqrt{\frac{r}{rs} - 1} \cdot e^{\frac{r}{2 \cdot rs}} \cdot \sinh\left(\frac{t}{2 \cdot rs}\right) \quad (10)$$

```
> U1 := (t, r) -> sqrt(r/rs-1)*exp(r/2/rs)*cosh(t/2/rs);
```

$$U1 := (t, r) \mapsto \sqrt{\frac{r}{rs} - 1} \cdot e^{\frac{r}{2 \cdot rs}} \cdot \cosh\left(\frac{t}{2 \cdot rs}\right) \quad (11)$$

```
> V2 := (t, r) -> sqrt(1-r/rs)*exp(r/2/rs)*cosh(t/2/rs);
```

$$V2 := (t, r) \mapsto \sqrt{1 - \frac{r}{rs}} \cdot e^{\frac{r}{2 \cdot rs}} \cdot \cosh\left(\frac{t}{2 \cdot rs}\right) \quad (12)$$

```
> U2 := (t, r) -> sqrt(1-r/rs)*exp(r/2/rs)*sinh(t/2/rs);
```

$$U2 := (t, r) \mapsto \sqrt{1 - \frac{r}{rs}} \cdot e^{\frac{r}{2 \cdot rs}} \cdot \sinh\left(\frac{t}{2 \cdot rs}\right) \quad (13)$$

```
> R := 4;
```

$$R := 4 \quad (14)$$

```
> p0 := plot([U2(t,0), V2(t,0), t=-15..15], scaling = constrained,
  color = black, thickness = 3, view = [-8..8, -8..8]):
> p1 := plot([U2(t,1), V2(t,1), t=-15..15], scaling = constrained,
  linestyle = dot):
> p2p := plot(U, U=-5..5, color = red, linestyle = dash, labels=[U,
  V]):
> p2n := plot(-U, U=-5..5, color = red, linestyle = dash):
> p3 := plot([U1(t,3), V1(t,3), t=-15..15], scaling = constrained,
  linestyle = dot):
> p4 := plot([U1(t,4), V1(t,4), t=-15..15], scaling = constrained,
  linestyle = dot):
> pp1 := plot([U1(tsch-eval(tsch, r=R), r), V1(tsch-eval(tsch, r=R),
  r), r=2.001..8]):
> pp2 := plot([U2(Re(tsch-eval(tsch, r=R)), r), V2(Re(tsch-eval
  (tsch, r=R)), r), r=0..1.999]):
> display([p0, p1, p2p, p2n, p3, p4, pp1, pp2], view =[-3..3, -3..3]
  );
```



In this diagram, constant Schwarzschild coordinate r=4, 3, and 1 are the dotted curves. The event horizon r=2 is the dashed line, and the singularity r=0 is the thick black curve. Using the Kruskal-Szekeres coordinates, we see that the red world line of a particle smoothly crosses the event horizon. It is unlike what we saw earlier using Schwarzschild coordinates.

If we compactify and rotate the Kruskal diagram, we get a Penrose diagram. See MTW Eqs. (34.2) and (34.3).

```
> T1 := (t, r) -> arctan(V1(t,r) + U1(t,r)) + arctan(V1(t,r) - U1(t,
  r));
```

$$T1 := (t, r) \mapsto \arctan(V1(t, r) + U1(t, r)) + \arctan(V1(t, r) - U1(t, r)) \tag{15}$$

```
> X1 := (t, r) -> arctan(V1(t,r) + U1(t,r)) - arctan(V1(t,r) - U1(t,
  r));
```

$$X1 := (t, r) \mapsto \arctan(V1(t, r) + U1(t, r)) - \arctan(V1(t, r) - U1(t, r)) \tag{16}$$

```
> T2 := (t, r) -> arctan(V2(t,r) + U2(t,r)) + arctan(V2(t,r) - U2(t,
  r));
```

$$T2 := (t, r) \mapsto \arctan(V2(t, r) + U2(t, r)) + \arctan(V2(t, r) - U2(t, r)) \tag{17}$$

```
> X2 := (t, r) -> arctan(V2(t,r) + U2(t,r)) - arctan(V2(t,r) - U2(t,
  r));
```

$$X2 := (t, r) \mapsto \arctan(V2(t, r) + U2(t, r)) - \arctan(V2(t, r) - U2(t, r)) \tag{18}$$

```
> pr99 := plot([X1(t, 99), T1(t, 99), t=-99..99], scaling =
  constrained, linestyle = dot):
> pr4 := plot([X1(t, 4), T1(t, 4), t=-15..15], scaling =
  constrained, linestyle = dot):
> pr3 := plot([X1(t, 3), T1(t, 3), t=-15..15], scaling =
  constrained, linestyle = dot):
> pr2p := plot(X, X=-Pi..Pi, color = red, linestyle = dash, labels=
  [X, T]):
> pr2n := plot(-X, X=-Pi..Pi, color = red, linestyle = dash):
> pr1 := plot([X2(t, 1), T2(t, 1), t=-15..15], scaling =
  constrained, linestyle = dot):
> pr0 := plot([X2(t, 0), T2(t, 0), t=-15..15], scaling =
  constrained, color = black, thickness = 3):
> prp1 := plot([X1(tsch-eval(tsch, r=R), r), T1(tsch-eval(tsch, r=
```

```
R), r), r=2.001..8]):
> prp2 := plot([X2(Re(tsch-eval(tsch, r=R)), r), T2(Re(tsch-eval(tsch, r=R)), r), r=0..1.999]):
> display([pr99, pr4, pr3, pr2p, pr2n, pr1, pr0, prp1, prp2], view = [-Pi..Pi, -Pi/2..Pi/2]);
```



In this Penrose diagram, the dotted curves are r=99, 4, 3, and 1. The event horizon r=2 is the dashed lines, and the singularity r=0 is the thick black line. Again we see that the world line of the astronaut crosses the event horizon smoothly. In both Kruskal and Penrose diagrams, the light cone is defined by 45-degree lines, and a particle moves inside the light cone. Using the diagrams, we have a clear picture that once the astronaut passes the event horizon, he or she must end up at the physical singularity r=0, indicated by the thick black line.

We use Maple's DifferentialGeometry and Tensor packages to calculate the Riemann curvature tensor and the Kretschmann scalar, to show the nonsingularity of r=2*M.

```
> restart:
> with(DifferentialGeometry): with(Tensor):
> DGsetup([t, r, theta, phi], M);
```

frame name: M **(19)**

```
M > rs := 2*G*M;
```

$$rs := 2\,G\,M \quad (20)$$

```
M > g := evalDG(-(1-rs/r)*dt &t dt + 1/(1-rs/r)*dr &t dr + r^2*dtheta &t dtheta + r^2*sin(theta)^2*dphi &t dphi);
```

$$g := \frac{(2\,G\,M - r)\,dt}{r}\,dt - \frac{r\,dr}{2\,G\,M - r}\,dr + r^2\,d\theta\,d\theta + r^2\sin(\theta)^2\,d\phi\,d\phi \quad (21)$$

```
M > R := CurvatureTensor(g);
```

$$R := -\frac{2\,G\,M\,D\_t}{r^2\,(2\,G\,M - r)}\,dr\,dt\,dr + \frac{2\,G\,M\,D\_t}{r^2\,(2\,G\,M - r)}\,dr\,dr\,dt - \frac{G\,M\,D\_t}{r}\,d\theta\,dt\,d\theta \quad (22)$$
$$+ \frac{G\,M\,D\_t}{r}\,d\theta\,d\theta\,dt - \frac{G\,M\sin(\theta)^2\,D\_t}{r}\,d\phi\,dt\,d\phi + \frac{G\,M\sin(\theta)^2\,D\_t}{r}\,d\phi\,d\phi\,dt$$
$$- \frac{2\,(2\,G\,M - r)\,G\,M\,D\_r}{r^4}\,dt\,dt\,dr + \frac{2\,(2\,G\,M - r)\,G\,M\,D\_r}{r^4}\,dt\,dr\,dt$$

$$- \frac{G M D\_r}{r} d\theta\, dr\, d\theta + \frac{G M D\_r}{r} d\theta\, d\theta\, dr - \frac{G M \sin(\theta)^2 D\_r}{r} d\phi\, dr\, d\phi$$

$$+ \frac{G M \sin(\theta)^2 D\_r}{r} d\phi\, d\phi\, dr + \frac{(2 G M - r) G M D\_\theta}{r^4} dt\, dt\, d\theta$$

$$- \frac{(2 G M - r) G M D\_\theta}{r^4} dt\, d\theta\, dt - \frac{G M D\_\theta}{r^2 (2 G M - r)} dr\, dr\, d\theta + \frac{G M D\_\theta}{r^2 (2 G M - r)} dr\, d\theta\, dr$$

$$+ \frac{2 G M \sin(\theta)^2 D\_\theta}{r} d\phi\, d\theta\, d\phi - \frac{2 G M \sin(\theta)^2 D\_\theta}{r} d\phi\, d\phi\, d\theta$$

$$+ \frac{(2 G M - r) G M D\_\phi}{r^4} dt\, dt\, d\phi - \frac{(2 G M - r) G M D\_\phi}{r^4} dt\, d\phi\, dt$$

$$- \frac{G M D\_\phi}{r^2 (2 G M - r)} dr\, dr\, d\phi + \frac{G M D\_\phi}{r^2 (2 G M - r)} dr\, d\phi\, dr - \frac{2 G M D\_\phi}{r} d\theta\, d\theta\, d\phi$$

$$+ \frac{2 G M D\_\phi}{r} d\theta\, d\phi\, d\theta$$

```
M > RL := RaiseLowerIndices(g, R, [1]);
```

$$RL := - \frac{2 G M dt}{r^3} dr\, dt\, dr + \frac{2 G M dt}{r^3} dr\, dr\, dt - \frac{(2 G M - r) G M dt}{r^2} d\theta\, dt\, d\theta \quad \textbf{(23)}$$

$$+ \frac{(2 G M - r) G M dt}{r^2} d\theta\, d\theta\, dt - \frac{(2 G M - r) G M \sin(\theta)^2 dt}{r^2} d\phi\, dt\, d\phi$$

$$+ \frac{(2 G M - r) G M \sin(\theta)^2 dt}{r^2} d\phi\, d\phi\, dt + \frac{2 G M dr}{r^3} dt\, dt\, dr - \frac{2 G M dr}{r^3} dt\, dr\, dt$$

$$+ \frac{G M dr}{2 G M - r} d\theta\, dr\, d\theta - \frac{G M dr}{2 G M - r} d\theta\, d\theta\, dr + \frac{G M \sin(\theta)^2 dr}{2 G M - r} d\phi\, dr\, d\phi$$

$$- \frac{G M \sin(\theta)^2 dr}{2 G M - r} d\phi\, d\phi\, dr + \frac{(2 G M - r) G M d\theta}{r^2} dt\, dt\, d\theta$$

$$- \frac{(2 G M - r) G M d\theta}{r^2} dt\, d\theta\, dt - \frac{G M d\theta}{2 G M - r} dr\, dr\, d\theta + \frac{G M d\theta}{2 G M - r} dr\, d\theta\, dr$$

$$+ 2 r G M \sin(\theta)^2 d\theta\, d\phi\, d\theta\, d\phi - 2 r G M \sin(\theta)^2 d\theta\, d\phi\, d\phi\, d\theta$$

$$+ \frac{(2 G M - r) G M \sin(\theta)^2 d\phi}{r^2} dt\, dt\, d\phi - \frac{(2 G M - r) G M \sin(\theta)^2 d\phi}{r^2} dt\, d\phi\, dt$$

$$- \frac{G M \sin(\theta)^2 d\phi}{2 G M - r} dr\, dr\, d\phi + \frac{G M \sin(\theta)^2 d\phi}{2 G M - r} dr\, d\phi\, dr - 2 r G M \sin(\theta)^2 d\phi\, d\theta\, d\theta\, d\phi$$

$$+ 2 r G M \sin(\theta)^2 d\phi\, d\theta\, d\phi\, d\theta$$

```
M > RU := RaiseLowerIndices(InverseMetric(g), R, [2,3,4]);
```

$$RU := - \frac{2 G M D\_t}{r^3} D\_r D\_t D\_r + \frac{2 G M D\_t}{r^3} D\_r D\_r D\_t - \frac{G M D\_t}{r^4 (2 G M - r)} D\_\theta D\_t D\_\theta \quad \textbf{(24)}$$

$$+\frac{G M D\_t}{r^4 (2 G M - r)} D\_\theta D\_\theta D\_t - \frac{G M D\_t}{r^4 (2 G M - r) \sin(\theta)^2} D\_\phi D\_t D\_\phi$$
$$+\frac{G M D\_t}{r^4 (2 G M - r) \sin(\theta)^2} D\_\phi D\_\phi D\_t + \frac{2 G M D\_r}{r^3} D\_t D\_t D\_r$$
$$-\frac{2 G M D\_r}{r^3} D\_t D\_r D\_t + \frac{(2 G M - r) G M D\_r}{r^6} D\_\theta D\_r D\_\theta$$
$$-\frac{(2 G M - r) G M D\_r}{r^6} D\_\theta D\_\theta D\_r + \frac{(2 G M - r) G M D\_r}{r^6 \sin(\theta)^2} D\_\phi D\_r D\_\phi$$
$$-\frac{(2 G M - r) G M D\_r}{r^6 \sin(\theta)^2} D\_\phi D\_\phi D\_r + \frac{G M D\_\theta}{r^4 (2 G M - r)} D\_t D\_t D\_\theta$$
$$-\frac{G M D\_\theta}{r^4 (2 G M - r)} D\_t D\_\theta D\_t - \frac{(2 G M - r) G M D\_\theta}{r^6} D\_r D\_r D\_\theta$$
$$+\frac{(2 G M - r) G M D\_\theta}{r^6} D\_r D\_\theta D\_r + \frac{2 G M D\_\theta}{r^7 \sin(\theta)^2} D\_\phi D\_\theta D\_\phi$$
$$-\frac{2 G M D\_\theta}{r^7 \sin(\theta)^2} D\_\phi D\_\phi D\_\theta + \frac{G M D\_\phi}{r^4 (2 G M - r) \sin(\theta)^2} D\_t D\_t D\_\phi$$
$$-\frac{G M D\_\phi}{r^4 (2 G M - r) \sin(\theta)^2} D\_t D\_\phi D\_t - \frac{(2 G M - r) G M D\_\phi}{r^6 \sin(\theta)^2} D\_r D\_r D\_\phi$$
$$+\frac{(2 G M - r) G M D\_\phi}{r^6 \sin(\theta)^2} D\_r D\_\phi D\_r - \frac{2 G M D\_\phi}{r^7 \sin(\theta)^2} D\_\theta D\_\theta D\_\phi$$
$$+\frac{2 G M D\_\phi}{r^7 \sin(\theta)^2} D\_\theta D\_\phi D\_\theta$$

```
M > ContractIndices(RL, RU, [[1,1],[2,2],[3,3],[4,4]]);
```

$$\frac{48 G^2 M^2}{r^6} \quad \textbf{(25)}$$

See MTW Eq. (31.7). This scalar quantity behaves perfectly at the event horizon r=rs=2*M. It is only singular when r=0.

References: C. W. Misner, K. S. Thorne and J. A. Wheeler, Gravitation, Freeman, 1973.